

BASIC

Bally

programmed instruction course

THE SELF-TEACHING SYSTEM
FOR MAKING YOUR OWN
COMPUTER GAMES, ELECTRONIC
MUSIC, AND VIDEO ART

Bally BASIC

BASIC is a language designed to make computers easy to understand and use. Simple words like RUN and PRINT tell your computer what to do.

There are many versions of BASIC as well as several other computer languages. Palo Alto Tiny Basic, developed by Lichen Wang, eliminates many complex expressions used in mathematics and physics and is particularly easy for beginners to learn.

Bally BASIC, written by Jay Fenton, is an expanded version of Palo Alto Tiny Basic that allows you to draw pictures, select colors, and play music on your TV. By adding full color graphics and sound, Bally BASIC expands your Bally Professional Arcade to include colorful computer games, electronic music, and video art.

This programmed instruction course, written by Dick Ainsworth, is your introduction to understanding and using Bally BASIC. You will learn how to talk to your computer in a few minutes and then you can expand your knowledge and enjoyment in the directions that interest you most.

Published by Bally Consumer Products Division
Bally Manufacturing Corporation
10750 West Grand Ave., Franklin Park, Ill. 60131

First Edition April, 1978

©Bally Manufacturing Corp. 1978 All rights reserved.

INDEX

Operating Instructions 4

Programming Course

| | |
|---|----|
| Lesson 1. Printing, counting and loops | 18 |
| Lesson 2. Random numbers, inputs and what if? | 30 |
| Lesson 3. Subroutines | 40 |
| Lesson 4. Strings | 48 |
| Lesson 5. Electronic Music | 54 |
| Lesson 6. Graphics | 64 |
| Lesson 7. Computer Games | 78 |
| Lesson 8. Video Art | 90 |

Programs 98

Computer Games
Electronic Music
Graphs and charts
Video Art
Learning Skills

Terms and Symbols 122


Computer Words
Inputs, outputs and controls
Error messages

Operating Instructions

If you are using your Bally Professional Arcade for the first time, please follow the directions in the *Owner's Manual* packaged with your unit. Connect your Bally Professional Arcade to a black and white or color TV and try out several of the games.

After you are familiar with your arcade and know how it operates, try Bally BASIC and discover the enjoyment of having your own personal computer.

REMOVE the keypad overlay from its envelope in the front of this manual. (This envelope is a good place to store your overlay when you're not using it.)



INSERT the Bally BASIC cassette in the cassette slot and press down firmly.



PLACE the keypad overlay on the keypad.

RESET your computer by pressing the **RESET** button next to the cassette. Your TV screen should look like this picture.



CAUTION

RESET erases your program. If you press this button by accident you must enter your program again, from the beginning.


EJECT causes your programming cassette to pop up so you can remove it. Pressing the eject button accidentally will cause your program to stop.

If this happens, push the cassette back into place, press **RESET**, and enter your program again.

STATIC

The same static that causes dots on your TV screen or noise in the speaker can effect your computer. If static interrupts your program and causes it to stop, press **RESET** and enter it again.


Numbers



Your Bally BASIC keypad is divided into three separate kinds of information: NUMBERS, LETTERS AND WORDS.

The WHITE numbers and symbols on your keypad are printed on your TV screen when you push those keys.

1
2
3
SPACE
4




Now use the ERASE key to remove the numbers from the screen.

ERASE
ERASE
ERASE
ERASE
ERASE



Letters



The GREEN shift key selects characters on the left.

The RED shift key selects characters in the center.

The BLUE shift key selects characters on the right.

To print a letter or character on your TV screen, use the shift key in the same color. First press either the GREEN, RED or BLUE shift key to select the color of the letter you want. Then press the key that is under the letter you want to print.

RED

9

BLUE

9

SPACE

RED

6

RED

9

RED

8

BLUE

5

RED


8

BLUE

1



Words



The GOLD shift key selects the WORDS printed in GOLD.

Words the computer understands are printed on the keypad in GOLD. Press the WORDS key, and then press the key under the words you want to print on your TV.

WORDS

BOX

WORDS

STEP

WORDS

FOR



You can now print numbers, letters and words on the screen. Next you will learn to put programs into your computer.

Reset the computer by pressing the RESET button.

RESET



The RESET button erases all instructions and programs in the computer's memory and clears the screen.

Now you will enter a short program.

Number the first instruction 10. Use the WORDS key to say PRINT and then spell out "HELLO!"

10PRINT "HELLO!"

GO



The GO key acts like a carriage return on a typewriter and moves you to the next line.

Now add the second instruction to your program and number it 20. Notice that GOTO is one word. Press the GO key to end the line.

20GOTO 10

GO



Now the program is in the computer memory. To look at the complete program, ask the computer to LIST it.

LIST

GO




Check your program and see if it matches the example. If your TV screen doesn't match this picture, RESET your computer and enter line number 10 and line number 20 again.

Now you can run the program. The computer will print the word "HELLO!" Then it will go back to the beginning of your program and start over. To stop the program, press the halt key, **H** and hold it down until the computer halts.

RUN


GO



Now LIST your program again.

LIST

GO



You need to know how to change a program so that you can fix a mistake or make your program do something different. To change an instruction, just enter the number of the line you want to change and then enter the new instruction.

Now enter this new instruction to replace line 10 in your program. Don't forget the comma at the end!

10PRINT "BYE!",

GO

LIST

GO



```
HELLO!  
BALLY BASIC  
>LIST  
10 PRINT "HELLO!"  
20 GOTO 10  
>10PRINT "BYE!",  
>LIST  
10 PRINT "BYE!",  
20 GOTO 10  
>
```

If your program matches the example, run it and see what it does. Use the halt key **H** to stop.

RUN

GO

H



```
E!BYE!BYE!BYE!BYE!BYE!BYE!  
BYE!BYE!BYE!BYE!BYE!BYE!  
E!BYE!BYE!BYE!BYE!BYE!BYE!  
BYE!BYE!BYE!BYE!BYE!BYE!  
E!BYE!BYE!BYE!BYE!BYE!BYE!  
BYE!BYE!BYE!BYE!BYE!BYE!  
E!BYE!BYE!BYE!BYE!BYE!BYE!  
BYE!BYE!BYE!BYE!BYE!BYE!  
E!BYE!BYE!BYE!BYE!BYE!BYE!  
BYE!BYE!BYE!BYE!BYE!BYE!  
E!BYE!BYE!BYE!BYE!BYE!  
BALLY BASIC  
>
```

Now enter and run this longer program. Your computer will number the lines 10, 20, 30, and 40 automatically. After RESET and at the end of each line, press the WORDS key before you press the GO key.

RESET

10X=RND(50)+20

20Y=RND(80)-40


30LINE X,Y,3

40GOTO 10

GO

LIST


GO



Check your program to make sure it matches the example, then run it.

RUN

GO



Press the PAUSE key to stop the program. Press any key to start it again.

With this feature you can also pause while listing long programs.

In the Programs Section are many programs you can select from. Remember to RESET before each program and to press GO after each line. If you want to number lines automatically press WORDS and GO.

REVIEW

Now make sure you understand how to operate your computer and enter and run programs.

1. Insert your Bally BASIC Programming Cassette and put the keypad overlay in place.
2. Press RESET (next to cassette). This erases any old programs.
3. Enter each instruction and press GO. Or press WORDS and GO for a new line number.
4. LIST the program, and check each instruction carefully. PAUSE key lets you pause when listing long programs.
5. If there are any mistakes enter the instruction again using the same line number. To remove an instruction completely re-enter its line number and press GO.
6. When your program matches the example press RUN and GO.

Now you have two choices. You can go to LESSON ONE and continue learning how to write your own programs, or you can go to the PROGRAMS section of this manual and try out any of the programs you like.

Programming Course

Lesson 1 Printing, counting and loops.

Before you begin these lessons please read and understand the OPERATING INSTRUCTIONS. They begin on page 4 and show you how to enter, list and run programs on your computer.

Learning to write your own programs isn't hard at all. Soon you will be able to have your computer play your own games, music and video art.

Let's begin by writing a short program.

First RESET the computer with the RESET button:


RESET



Now spell out HI and your name. Press GO to end the line.

HI BILL

GO




The computer is saying WHAT? because it doesn't know what you said. The words HI and BILL are not words your computer understands.

Instead, try it this way: use the WORDS shift key to enter the word PRINT. Then spell out "HI BILL". Don't forget the quotation marks.

PRINT "HI BILL"


GO



PRINT is one of the special words your computer understands. When you pressed GO, the computer followed your instruction and printed the words between the quotation marks.

Now press GO again and see what happens:

GO



```
BILLY BASIC
>HI BILL!

WHAT?
BILLY BASIC
>PRINT "HI BILL"
HI BILL
>
```

You can't print these words a second time because the computer doesn't remember what to do. To have your computer remember your instruction, just give it a line number.

Number your instruction 10 and enter it again.

10PRINT "HI BILL"

GO




```
BILLY BASIC
>HI BILL!

WHAT?
BILLY BASIC
>PRINT "HI BILL"
HI BILL
>
10PRINT "HI BILL"
>
```

Now you have a one-line program in the computer memory. You can run this program as many times as you like.

To run your program, use the WORDS shift key and enter RUN.

RUN
GO




```
>HI BILL
WHAT?
BILLY BASIC
>PRINT "HI BILL"
HI BILL
>
>10PRINT "HI BILL"
>RUN
HI BILL
>
```

Add a second instruction to your program and number it 20.

LIST your program.

20GOTO 10
GO
LIST
GO




```
>PRINT "HI BILL"
HI BILL
>10PRINT "HI BILL"
>RUN
HI BILL
>20GOTO 10
>LIST
10 PRINT "HI BILL"
20 GOTO 10
>
```

Here's what your new program will do. The computer will print HI BILL , go back to the beginning of your program, print HI BILL , go back to the beginning again, print HI BILL , ...on and on until you press the HALT key.

RUN

GO



Press and hold halt  until the computer stops.

How many times did you run your program? There's an easy way to find out. Make a counter to keep track of the number of items it ran.

RESET your computer, then enter and LIST this new program.

RESET

10A=0

20A=A+1


30PRINT "HI BILL"

40GOTO 20

GO

LIST

GO



This program uses the letter A as a counter. Here's what happens when you run it.

In line 10 the computer puts a zero in the A counter.


In line 20 the computer adds 1 to the A counter.

In line 30 the computer prints whatever is between the quotation marks.

In line 40 the computer goes back to line 20, adds one more to the A counter, and repeats.


RUN your program and print HI BILL about a dozen times. Then press and hold the halt **H** key.

RUN
GO



Each time the computer printed HI BILL it added 1 to the A counter. To find out how many times your program ran, see what number is in the A counter.

PRINT A
GO



In the example shown here the program ran 12 times and $A = 12$.

When you say print "A" the computer prints the letter A. When you say PRINT A the computer prints the number in the A counter.

You can also use any other letter you want to be a counter.

Now LIST your program again.

LIST
GO

```
HI BILL
BILLY BASIC
>PRINT A
12
>LIST
10 A=0
20 A=A+1
30 PRINT "HI BILL"
40 GOTO 20
>■
```


Instead of printing HI BILL over and over, you can print the number in the A counter. Change line 30 by entering the same line number and adding the new instruction, then LIST your program.

30PRINT A
GO
LIST
GO

```
10 A=0
20 A=A+1
30 PRINT "HI BILL"
40 GOTO 20
>30PRINT A
>LIST
10 A=0
20 A=A+1
30 PRINT A
40 GOTO 20
>■
```


The computer always lists the latest version of your program. This time before you run your program, try to figure out what it will do. Now RUN it and see if you were right.

RUN
GO



The computer puts a zero in the A counter in line 10. In line 20, 1 is added to A. Next, in line 30, the computer prints the number A. Then the computer loops back to line 20 and repeats.

COUNTING LOOP

```
10 A=0
20 A=A+1
30 PRINT A
40 GOTO 20
```

WHAT THE COMPUTER DOES

```
10 A=0
20 A=1
30 PRINT 1
40 GOTO 20
```

```
20 A=2
30 PRINT 2
40 GOTO 20
```

```
20 A=3
30 PRINT 3
40 GOTO 20
```

until you press halt H

Programs that repeat are called loops. Another way to program a loop is with the words FOR and NEXT. RESET your computer to erase the counting loop and enter this program.

RESET

10FOR A=1TO 12


20PRINT A

30NEXT A

GO

LIST

GO




In line 10 the computer puts 1 in the A counter. In line 20 A is printed.

The word NEXT in line 30 means add 1 to A and loop back to word FOR. NEXT A replaces $A=A+1$ and GOTO 20 which were used in the last program.

Now RUN your program and print the number in A as the A counter goes from 1 to 12.


RUN

GO



This time the program loop stopped automatically at 12. LIST your program again.

LIST
GO



```
7
8
9
10
11
12
>XLIST
10 FOR A=1 TO 12
20 PRINT A
30 NEXT A
>
```


FOR/NEXT LOOP
10 FOR A=1 TO 12
20 PRINT A
30 NEXT A

WHAT THE COMPUTER DOES

10 A=1
20 PRINT "1"
30 A=2;GOTO 20
20 PRINT 2
30 A=3;GOTO 20
20 PRINT 3
30 A=4;GOTO 20
until A=12

The FOR/NEXT loop adds 1 to the counter. You can also add 2, 3, or any other number. Change line 10 to count by 2's.

10FOR A=1 TO 12 STEP 2
GO
LIST
GO




```
12
>XLIST
10 FOR A=1 TO 12
20 PRINT A
30 NEXT A
>XLIST
10 FOR A=1 TO 12 STEP 2
20 PRINT A
30 NEXT A
>
```

Now RUN your program and see if it prints all the odd numbers between 1 and 12.

RUN

GO



You could also change line 10 and print all the tens from one to one hundred or all the leap years since your birthday. You can even step backwards by using negative numbers. RESET and enter this new program.

RESET

10FOR X=10 TO 0 STEP-1

20PRINT X


30NEXT X

40PRINT "BLAST OFF!"

GO

LIST


GO



Now RUN your program. You're at 10 seconds and counting!

RUN

GO




Now for some fun to end your first lesson.

Add these three lines to your program:


```
50FC=7
60BC=0
70PRINT "10 30 50 80 0
    0 50 80000"
```

```
GO
LIST
GO
```



Lines 50 and 60 change the colors each time your program runs and line 70 plays a tune at the end. Try it!

```
RUN
GO
```



You will learn all about colors in lesson 8, and music is explained in lesson 5. The remaining lessons are no more difficult than the one you have just completed.

Now go on to lesson two—or skip ahead to any other lesson you would like.

Programming Course

Lesson 2

Random numbers, inputs, and what if?

It's often handy to have your computer pick out numbers at random. Here's a program that selects random numbers between one and twenty.

RESET

10A=RND (20)


20PRINT A

30GOTO 10

GO

LIST


GO



In line 10 the computer will make the A counter equal to a random number between one and twenty. In line 20 the computer prints the number in A. Line 30 sends the computer back to line 10. The computer continues picking a random number, printing it, and looping back to the beginning of the program.


The numbers this program selects are different each time, so don't expect your numbers to match the example.

RUN
GO




Now change line 10 to put random numbers from one to three in the A counter.

H
10A=RND (3)
GO
LIST
GO



Now RUN your program and let it list a few numbers.

RUN
H



IF is a computer word that lets you check and see whether something is true or not.

Enter this program:


RESET

```
10A=RND (2)
20IF A=1 PRINT "HEADS"
30IF A=2 PRINT "TAILS"
40GOTO 10
```

GO

LIST

GO



```
>
>10A=RND (2)
>20IF A=1 PRINT "HEADS"
>30IF A=2 PRINT "TAILS"
>40GOTO 10
>LIST
10 A=RND (2)
20 IF A=1 PRINT "HEADS"
30 IF A=2 PRINT "TAILS"
40 GOTO 10
>■
```

First the computer makes the A counter either 1 or 2. Then if A=1 the computer prints "HEADS" and if A=2 the computer prints "TAILS". Then the computer goes back to line 10 and again sets the A counter to either 1 or 2, and the loop continues.

The computer is using RND (2) to change the number in the A counter. Depending on whether the number is 1 or 2, the computer prints either "HEADS" or "TAILS". Now run the program and see if heads or tails come up more often.

RUN

GO



```
TAILS
HEADS
HEADS
TAILS
TAILS
TAILS
TAILS
HEADS
TAILS
HEADS
■
```


Another way to change numbers in a program is to enter them yourself with INPUT A.

When the computer reads INPUT A, it waits for you to enter a number before it continues running the program.

The letter A after the word INPUT tells the computer which counter to use. In this program the number you input will be stored in the A counter.

```
RESET  
10INPUT A  
20PRINT A  
30GOTO 10  
GO  
LIST  
GO
```




When you run the program, the computer will stop and wait for you to input a number.

After you input a number, the computer prints the number you entered and asks for a second number.

Follow the suggestions below or try your own.

4
GO
23
GO
731
GO




The computer prints A to remind you that your input will be stored in the A counter.

You can have the computer remind you in other ways, too.

Try this change in the program:

H
10 INPUT "YOUR NUMBER?" A
GO
LIST
GO
RUN
GO



Now enter these numbers:

27


GO

2345678

GO

21

GO



The largest number your computer's memory can hold is 32767. You just saw what happens when you input a number larger than that.

The computer will ask WHAT? when it doesn't understand you.

It will ask HOW? when it understands but can't do what you requested.

You have been using INPUT A to put numbers in the A counter. This program inputs numbers into two counters and then prints their sum.

RESET

10INPUT A

20INPUT B


30 PRINT A+B

40GOTO 10

GO

LIST


GO



The first number you INPUT goes in the A counter, and the second number goes in the B counter. The computer prints their sum $A+B$ and loops back to the beginning of your program.

Try adding these numbers together then try some of your own.


```
RUN
GO
2
GO
3
GO
```



```
>40GOTO 10
>LIST
10 INPUT A
20 INPUT B
30 PRINT A+B
40 GOTO 10
>RUN
A 2
B 3
5
A █
```

INPUT lets you put numbers into the computer and RND has the computer pick numbers at random. Now you can combine these and build a guessing game.

```
RESET
10A=RND (10)
20INPUT "YOUR GUESS:"B
30IF A=B GOTO 70
40IF A>B PRINT "MORE↑"
50IF A<B PRINT "LESS↓"
60GOTO 20
70PRINT B, "IS RIGHT!"
80GOTO 10
GO
LIST
GO
```



```
>80GOTO 10
>LIST
10 A=RND (10)
20 INPUT "YOUR GUESS:"B
30 IF A=B GOTO 70
40 IF A>B PRINT "MORE↑"
50 IF A<B PRINT "LESS↓"
60 GOTO 20
70 PRINT B, " IS RIGHT!"
80 GOTO 10
>█
```

This program is longer than your others so we'll look at it step-by-step.

First the computer picks a random number between one and ten and stores it in the A counter. Then you try to guess the number, and your input is stored in the B counter.

Now there are three things that can be true. If $A=B$ then your guess is right. The computer goes to line 70 and prints your answer and the words IS RIGHT! If A is larger than B, $A>B$, then your guess is too small. The computer prints MORE. If A is less than B, $A<B$, your guess is too big and the computer prints LESS.

There are two loops in this program. If $A=B$ the computer goes to line 70, prints the number you picked and the words IS RIGHT! and then loops back to the beginning to start a new game.

If you didn't get the right answer the computer loops back to line 20 so you can try again.

RUN

GO

2

GO

4

GO

5


GO

7

GO

6

GO



Here's how a sample run might look.


You can change line 10 to `A=RND (100)` and make the game harder, or add a counter to keep track of the number of guesses it took. Any of the words inside the quotation marks, like "MORE" can be changed to say whatever you want.

Before you try your game on your friends, learn how to win every time. When the computer asks for your guess, just enter the letter A.

Here's a program add-on that you will like. Just enter the line numbers as shown and the computer will put your instructions in the right order.




```
80NT=20
90PRINT "60605-5-504-5-50"
100NT=3
110GOTO 10
GO
LIST
```



Now try the guessing game again and be ready for a surprise when you get the answer right!

RUN
GO



Programming Course

Lesson 3 Subroutines


In writing longer programs you may want to use a shortcut. GOSUB and RETURN make it easy to use the same instruction several places in your program.

This program prints the words ROCK, SHEARS and PAPER several times. To avoid having to write these same instructions over and over, we will use GOSUB and RETURN.

Enter the first part of your program.

```
RESET
10GOSUB 201
20GOSUB 202
30GOSUB 203
40GOTO 10
201PRINT "ROCK";RETURN
202PRINT "SHEARS";RETURN
203PRINT "PAPER";RETURN
```

```
GO
LIST
GO
```



Here's what's going to happen. When the computer reads line 10, it will jump to line 201 and continue until it reaches the word RETURN. Then the computer will jump back to line 10 and continue.

The same thing will happen in lines 20 and 30. The computer will jump to the GOSUB instructions and then return.

In line 40 the GOTO instruction tells the computer to go back to line 10 and start the program over again.

Now RUN this part of your program.

RUN
GO



Now add these additional lines to your program.


Your program is now too long to fit on the TV screen. Press and hold the PAUSE key to stop the listing at line 90 so you can check it. Press GO to continue the listing.

```
40A=RND (3)
50INPUT "YOUR CHOICE:"B
60PRINT "I PICKED ",
70GOSUB 200+A
80PRINT "YOU PICKED",
90GOSUB 200+B
100GOTO 10
```

GO

LIST

GO



After you pause at line 90 and check your list. Then press GO to finish list.

GO



Here's what you've added.

In line 40 the computer will select 1, 2, or 3 at random and put this number in the A counter. In line 50 the computer will ask for your choice (1, 2 or 3) and the number you INPUT will go into the B counter.


At line 60, the computer will print I PICKED and at line 70 it will GOSUB to line number 200+A.

If A=1, the computer will GOSUB to line 201. If A=2, it will GOSUB to 202. And if A=3, it will GOSUB to 203. Depending on the value of the A counter, ROCK, SHEARS, or PAPER will be printed after the words I PICKED.

Lines 80 and 90 use the same GOSUB feature to print your selection. Line 100 loops the program back to the beginning.

Now RUN your program and INPUT 1, 2, or 3 to select ROCK, SHEARS or PAPER.

```
RUN
GO
1
GO
2
GO
3
GO
```



Now you can play ROCK, SHEARS, PAPER with your computer. The rules are:

ROCK breaks SHEARS

SHEARS cut PAPER

PAPER wraps ROCK

But let's have the computer tell us who won. HALT the program and add that feature with these lines.

H

```
100IF A=B PRINT "A TIE!";  
      GOTO 10  
110IF A=1 IF B=3GOTO 160  
120IF A=2 IF B=1GOTO 160  
130IF A=3 IF B=2GOTO 160  
140PRINT "I WIN!"  
150GOTO 10  
160PRINT "YOU WIN!"  
170GOTO 10
```


GO

LIST

GO

PAUSE will stop the list so you can check it.

GO



```
90 GOSUB 200+8  
100 IF A=B PRINT "A TIE!"  
;GOTO 10  
110 IF A=1 IF B=3GOTO 160  
120 IF A=2 IF B=1GOTO 160  
130 IF A=3 IF B=2GOTO 160  
140 PRINT "I WIN!"  
150 GOTO 10  
160 PRINT "YOU WIN!"  
170 GOTO 10
```



```
120 IF A=2 IF B=1GOTO 160  
130 IF A=3 IF B=2GOTO 160  
140 PRINT "I WIN!"  
150 GOTO 10  
160 PRINT "YOU WIN!"  
170 GOTO 10  
201 PRINT "ROCK";RET  
202 PRINT "SHEARS";RET  
203 PRINT "PAPER";RET
```

If you would like your computer to keep score, just add these lines. The computer will place them in your program automatically.



6H=0

8C=0

32PRINT " HUMAN:",H

34PRINT "COMPUTER:",C

145C=C+1

165H=H+1

If you want to add music, these instructions will do it.

141NT=10

142PRINT "135×105×10000"

143NT=3


161NT=10

162PRINT "3050034050000"

163NT=3

Now RUN your program and see if you can beat your computer.

RUN
GO



Here's a complete listing of your ROCK, SHEARS, PAPER game.

```
6 H=0
8 C=0
10 GOSUB 201
20 GOSUB 202
30 GOSUB 203
32 PRINT "HUMAN: ".H
34 PRINT "COMPUTER: ".C
40 A=RND (3)
50 INPUT "YOUR CHOICE: "B
60 PRINT "I PICKED ".
```

```
70 GOSUB 200+A
80 PRINT "YOU PICKED ".
90 GOSUB 200+B
100 IF A=B GOTO 10
110 IF A=1 IF B=3GOTO 160
120 IF A=2 IF B=1GOTO 160
130 IF A=3 IF B=2GOTO 160
140 PRINT "I WIN!"
141 NT=10
142 PRINT "135x105x10000"
```

```
143 NT=10
144 C=C+1
145 GOTO 10
146 PRINT "YOU WIN!"
147 NT=10
148 PRINT "3050034050000"
149 NT=3
150 H=H+1
170 GOTO 10
201 PRINT "ROCK";RETURN
```

```
202 PRINT "SHEARS";RETURN
203 PRINT "PAPER";RETURN
```


Programming Course

Lesson 4 Strings

It's often handy to be able to work with a sequence of numbers or letters. These are called strings or arrays and you can have a string of numbers, a string of letters or a string of musical notes.


Here's how strings work. The @ character is your computer's symbol for a string. The first item is AT location 1, or @(1), the second item in the string is AT location 2, or @(2), the third item is AT location 3, or @(3) and so on.

To find the number at location 4 in a string, you would ask for @(4) like this.

RESET


PRINT @(4)

GO



The fourth location in the string contains a zero. Store the number 12 at location 4 like this, then check it.


```
@(4)=12
GO
PRINT @(4)
GO
```



This program lists the numbers stored at the first ten locations in the @ string.


When you RESET the computer each location in the string is filled with a zero.

```
RESET
10FOR A=1TO 10
20PRINT A,@(A)
30NEXT A
GO
LIST
GO
```



As the A counter advances from 1 to 10, the computer prints 1 and then the number stored at the first location, 2 and the number stored at the second location and so on up to 10 and the tenth number stored in the string.

RUN
GO



Now enter these instructions. (Don't press WORDS before you press GO because no line numbers are needed.)

Each time you press GO the computer follows your instruction; it stores number 22 at string location 7, number 17 at string location 3, and 5 at string location 5.

@(7)=22


GO

@(3)=7

GO

@(5)=5


GO



Now RUN your program and see what numbers are stored at the first 10 string locations.

RUN

GO



This program prints a simple graph, using the string to store the numbers to be plotted.


RESET

```
10INPUT N
20FOR A=1 TO N
30INPUT @(A)
40NEXT A
50FOR A=1 TO N
60FOR B=1 TO @(A)
70PRINT #1, "S",
80NEXT B
90PRINT
100NEXT A
```

GO

LIST

GO




In line 10 you will set N equal to the number of items in the graph. The loop in lines 20, 30, and 40 stores the value of each item.

Lines 50 through 100 are a FOR/NEXT loop using the A counter. This loop prints each item in the graph.

Lines 60, 70, and 80 are a smaller loop that “counts” each item. For example, if @(3) is a 6, this loop will cycle six times and print \$\$\$\$\$\$ on the screen.

RUN your program and draw a bar graph. Enter the number of items, then the value of each item.

RUN
GO
5
GO
4
GO
7
GO
21
GO
13
GO
6
GO



In the next lesson you will see how strings can be used to store and play back musical notes.

Programming Course

Lesson 5 Electronic Music

There are two ways you can play music on your computer, MU and PRINT. This program sets MU equal to a random number between 31 and 87. Numbers in this range produce musical notes in your TV speaker.

Enter and RUN this random music generator.

```
RESET  
10MU=RND (57)+30  
20GOTO 10  
GO  
RUN  
GO
```



To change the speed of the notes adjust the built in note timer, NT. HALT your program and set the note time to 10.


H

NT=10

GO

RUN

GO

A screenshot of a computer screen displaying a BASIC program. The text on the screen is: BALLY BASIC, >, >10MU=RND (57)+30, >20GOTO 10, >RUN, BALLY BASIC, >NT=10, >RUN, and a small black square cursor at the end of the line. The screen has a thick black border.

```
BALLY BASIC
>
>10MU=RND (57)+30
>20GOTO 10
>RUN
BALLY BASIC
>NT=10
>RUN
■
```

With PRINT and the numbers 1 through 7 you can play a musical scale.

RESET

10PRINT "1234567"

GO

RUN

GO

A screenshot of a computer screen displaying a BASIC program. The text on the screen is: BALLY BASIC, >, >10PRINT "1234567", >RUN, 1234567, and a small black square cursor at the end of the line. The screen has a thick black border.

```
BALLY BASIC
>
>10PRINT "1234567"
>RUN
1234567
■
```

The note timer automatically returns to 3 whenever you RESET.

Here are the notes you just played:




To expand this scale one octave higher, just put a multiplication sign in front of each number.

20PRINT "x1x2x4x5x6x7"

GO

RUN

GO




Your program now plays these notes:



Now add the lowest octave and play your computer's full musical scale. Put the division sign in front of the numbers 1 through 7.


```
30PRINT "÷1÷2÷3÷4÷5
      ÷6÷7"
```

```
GO
RUN
GO
```



Your computer's complete musical scale is now:


$\div 1$ $\div 3$ $\div 5$ $\div 7$ 2 4 6 $\times 1$ $\times 3$ $\times 5$ $\times 7$
 $\div 2$ $\div 4$ $\div 6$ 1 3 5 7 $\times 2$ $\times 4$ $\times 6$



Sharps are selected by using an addition (plus) sign in front of the numbers and flats are selected with a subtraction (minus) sign.

```
40PRINT "+1+2+3+4+5
      +6+7"
50PRINT "-1-2-3-4-5
      -6-7"
```

GO
RUN
GO




Always put the sharp or flat sign in front of the octave sign, like this: $- \div 2$ or $+ \times 4$.

Now RESET the computer and play this tune. Slow the music down by making the note time equal to 20.

```
RESET
10PRINT "321233302220
      3550"
```

GO
NT=20
GO
RUN
GO



Rhythm can be added two ways. You can space between notes or add a 0, depending on the sound you want. Try these examples and hear the difference.

RESET

10PRINT "100101100+20220110-11000"

20PRINT "1 1 11 +2 22 11 -11 "


GO

NT=20

GO

RUN

GO



Notice that the notes hold or continue when you use a 0. The space key makes a rest. RUN this program again if you want to listen to the difference.

This next program combines everything you have learned. Notice how the space and the 0's set the rhythm.

RESET

**10PRINT "240567650310 40
22-22301÷60"**


GO

NT=12

GO

RUN

GO



Now build a player piano that stores an entire song and then plays it back. You will enter this program in two sections so it will be easier to check.

RESET

10CLEAR

20A=0

30K=KP

40IF K="PRINT" GOTO 120

50IF K="CLEAR" GOTO 10


60IF K=30A=A-1;GOTO 100

**70IF K="LINE" INPUT NT;
GOTO 30**

GO

LIST

GO



```
XLIST
10 CLEAR
20 A=0
30 K=KP
40 IF K="PRINT" GOTO 120
50 IF K="CLEAR" GOTO 10
60 IF K=30 A=A-1;GOTO 10
0
70 IF K="LINE" INPUT NT:
GOTO 30
>|
```

Compare your program with the example, correct any errors, and then enter the second section.

80A=A+1

90@(A)=K

100TV=K

110GOTO 30

120CLEAR

130FOR C=1TO A

140TV=@(C)

150NEXT C

160GOTO 30

GO

LIST

GO




```
GOTO 30
80 A=A+1
90 @(A)=K
100 TV=K
110 GOTO 30
120 CLEAR
130 FOR C=1TO A
140 TV=@(C)
150 NEXT C
160 GOTO 30
>|
```

Check your program carefully. When you RUN it the screen will go blank.


Enter a scale and play it back with the word PRINT.

RUN
GO
1234567
PRINT
PRINT



To change the note time, use the word LINE, enter the new note time and press GO.


LINE
10
GO



With this program the GO key is *only* used after you enter a new note time.

Play back at the new note time, using PRINT as before.

PRINT



The word CLEAR is used to clear the memory so you can enter a new song. With ERASE you can back up and change any or all of the notes.

Now enter this song. The numbers are shown here in groups of four because there are four beats to a measure. Enter the numbers in a continuous line. Do not press GO at the end of each line.

CLEAR

100÷5 I've been
1÷512 working on the
3000 rail-
1000 road.
4004 All the
1020 live-long
3000 day.
000■

100÷5 I've been
1÷512 work-ing on the
3000 rail-
1033 road. Just to
3020 pass the
2030 time a-
2000 way.
000■

■ = Space Key

2002 Can't you
+1232 hear the whistle
1000 blow-
÷5000 ing?
4044 Rise up so
1122 early in the
3000 morn.
000■

÷6000÷7 Can't you
1÷71÷6 hear the captain
÷5000 shout-
1000 ing.
3040 Di-na
3020 blow your
1000 horn.
000■

LINE

10

GO

PRINT

If you would like to know more about the Player Piano Program, LIST it and read the following section.

The A counter keeps track of how many notes are stored in the @ string.

After clearing the screen and setting the A counter to 0, the computer waits for you to enter a number on the keypad. The K counter is set to this number.

Next the computer checks to see if any words have been entered. If you enter PRINT the program goes to line 120 to play back the notes.

If you enter CLEAR the computer goes back to the beginning of the program and sets the A counter to 0. Key 30 is the erase key; and if this is pressed the A counter is reduced by one.

The word LINE is used in this program to input a new number for NT, the note time.

After checking to see if you have entered any special words, the computer adds one to the A counter. The new note is added to the @ string (line 90) and shows on the TV (line 100). GOTO 30 sends the computer back to wait for the next input from the keypad (line 30).

If PRINT is entered, the computer goes to line 120 and starts the playback process. The screen is cleared, and a FOR/NEXT loop is started. Remember that the A counter keeps track of how many notes there are. This part of the program (lines 130, 140 and 150) loops once for each note until all the notes have been written on the TV and played.



Programming Course

Lesson 6 Graphics

With only the words LINE and BOX you can draw an endless variety of graphs and graphic designs on your TV.

Here's how LINE works.



```
RESET  
10LINE 60,30,1  
GO  
RUN  
GO
```



Your TV screen is 159 dots wide and 87 dots high. Zero is in the center. When you run this program, the computer starts in the center of your screen and draws a line to a point that's 60 dots to the right of the center (60) and 30 dots up from the center (30).

Now add these instructions to clear the screen and draw the second line.

```
5CLEAR  
20LINE -50,20,1  
GO  
RUN  
GO
```



This time the computer moved to a point 50 dots to the left of center (-50) and 20 dots up from center (20) to draw the second line.



Now add this instruction.

```
30LINE -40,-30,1
```

```
GO
```

```
RUN
```

```
GO
```



Now the computer moves to a point that's 40 dots to the left of center (-40) and 30 dots down from center (-30).



Continue drawing in the lower right section of your screen with this instruction that means 40 to the right (40) and 20 down (-20).

40LINE 40,-20,1

GO

RUN


GO



LIST your program and check to see that you have all the instructions properly entered.

LIST

GO





Finally, draw a line back to the center (0,0) to complete your first graphic design.

```
50LINE 0,0,1
```

```
GO
```

```
RUN
```

```
GO
```



Now write a program that fills the screen with random lines.

RESET

10CLEAR

20X=RND (160)-81

30Y=RND (88)-45


40LINE X,Y,1

50GOTO 20

GO

LIST

GO




```
>20X=RND (160)-81
>30Y=RND (88)-45
>40LINE X,Y,1
>50GOTO 20
XLIST
10 CLEAR
20 X=RND (160)-81
30 Y=RND (88)-45
40 LINE X,Y,1
50 GOTO 20
>■
```

The computer selects random numbers for X and Y. Then it draws a LINE to the point on the TV screen that is X dots right or left of center and Y dots up or down. It loops back and picks a new X and Y position and then continues drawing.

RUN

GO



The number 1 after LINE means draw a *black* line. There are four kinds of lines you can make.

LINE X,Y,1 = Black
LINE X,Y,2 = White
LINE X,Y,3 = Reverse
LINE X,Y,4 = None

Change line 40 and find out what “reverse” lines are.


H

40LINE X,Y,3

GO


LIST

GO




RUN

GO



HALT your program, CLEAR the screen, and LIST your program.


H
CLEAR
GO
LIST
GO



Here's how the computer draws lines that match the size of your TV screen.


In line 20 the computer picks a number for X between -80 on the far left edge of the screen and 79 on the right edge of the screen.

In line 30 the computer selects a random number for Y that's between -44 on the bottom edge of your screen and 43 on the top edge of your screen.





Now change your program and create “reverse” boxes all over your screen. Also change the background color (BC).

```
BC=22
40BOX X,Y,12,30,3
GO
LIST
GO
```



The random numbers X and Y, position the box on the screen. The next two numbers, 12 and 30 tell the computer how many dots wide and tall to make the box. The last number, 3, reverses as before.

```
RUN
GO
```



Now make something different. Change the size of the boxes to look like the holes in an IBM card. Change the last number in line 40 to a 1, which will make all the boxes black. Add some computer music with line 50.



```
40BOX X,Y,2,5,1
```

```
50MU=RND(20)
```

```
60GOTO 20
```

```
GO
```


```
LIST
```

```
GO
```




```
RUN
```

```
GO
```



This next program draws a graph. First it asks how many numbers you have. Then it asks for each number. Finally it draws a graph that might look like this.



Enter and LIST this part of the program.

RESET

10CLEAR

20INPUT "←A→"A

30FOR N=1 TO A

40PRINT N,

50INPUT "?"@N)

60IF @N)> 87GOTO 40

70NEXT N

GO

LIST

GO

```
>60IF @N)>87 GOTO 40
>70NEXT N
>XLIST
10 CLEAR
20 INPUT "←A→"A
30 FOR N=1 TO A
40 PRINT N,
50 INPUT "?"@N)
60 IF @N)>87 GOTO 40
70 NEXT N
```

In line 20, the computer asks how many items will the graph have and then stores the answer in A.

The FOR/NEXT loop prints the number of each item, stores the value in the string @ (N), and checks to see if the value is over 87. If it is over 87 it will not fit on the TV screen and the computer goes back to line 40 for a new input.

RUN this portion of the program.

RUN

GO

3

GO

3

GO


21

GO

95

GO

13




Now add the final section that draws the graph.

```
80X=-80
90CLEAR
100LINE X,@(1)-44,0
110FOR N=1TO A
120LINE X,@(N)-44,1
130BOX X,-43,1,2,1
140X=X+150÷A-1
150NEXT N
```

GO


LIST

GO



To start drawing the graph (line 80), the computer sets $X = -80$ (the left edge of your screen), clears the screen, and places the starting point for the series of lines that make the graph.

The number $@(1) - 44$ is the vertical distance or number of dots above or below the center of the screen. For example, if the first number in the $@$ string is 0, then the computer subtracts 44 to place this point on the bottom of the graph.



There are three instructions (120, 130 and 140) in the last FOR/NEXT loop. These instructions are run once for each item in the graph.

In the line 120 the computer draws a line from the last point to the next point. Line 130 places a small dot at the bottom of the graph.

Line 140 changes the X counter to move each point on the graph a short distance to the right. The graph is 150 dots wide and this distance is divided equally.

RUN the program and draw a graph with these twelve figures. Don't forget to push GO after each number.

RUN

GO

12

15

21

28

35

42

28

35


28

39

49

68

79



Remember, no single entry can be larger than 87, and no decimal points are accepted.

Now use your graph drawing program to make a graph of your grocery expenses, your company sales, or your favorite stock.

Programming Course

Lesson 7 Video Games

In this lesson you will learn how to use the hand controls while you are running a program. You will also build a video target game and see how larger programs are made from several small programs.

First plug a hand control into the number 1 socket (next to the power cord), and then enter this program.

RESET



10PRINT JX(1),JY(1)

20GOTO 10

GO


LIST

GO




With the number 1 hand control centered, JX(1) and JY(1) are zero. Moving the knob to the right makes JX(1)=1, and moving it to the left makes JX(1)=-1. Similarly, moving the knob forward or back makes JY(1) either 1 or -1. Run the program and change the numbers on your screen by moving the knob left and right, back and forth. Turning (rotating) the knob has no effect right now.

RUN
GO



Now use the hand control to move a box on the screen with this program. Two counters (X and Y) keep track of where the box is. When you move the box with the hand controls you will be adding 1 or -1 to the counters. RUN the program and move the box.

RESET
10X=X+JX(1)
20Y=Y+JY(1)
30BOX X,Y,10,10,1
40GOTO 10
GO
RUN
GO



The trigger is called TR(1) and TR(1)=1 when the trigger is pulled. Add this line to your program so you can clear the screen by pulling the trigger.


H

40IF TR(1)=1 CLEAR
50GOTO 10

GO

LIST


GO



Now RUN your program, draw some lines, and CLEAR the screen with the trigger.

RUN



GO



Now change your program and see what happens when you turn the knob. The knob on hand control number 1 is called KN(1).

RUN the program and turn the knob.

```
RESET  
10PRINT KN(1)  
20GOTO 10  
GO  
RUN  
GO
```



With the knob all the way to the left, $KN(1) = -128$ and with the knob turned to the right $KN(1) = 127$.

Try to dial your age. This is hard to do because the numbers are very close together on the knob.

This program spreads the numbers out and makes it easier to dial your age.

RESET


10PRINT $KN(1) \div 3 + 42$

20GOTO 10


GO

LIST

GO




Here's what you have done to make it easier. $KN(1)$ still has a range from -128 to 127 . When you divide $KN(1)$ by 3 this range is reduced to -42 on the left and 42 on the right. When the computer adds 42 to $KN(1) \div 3$ the final range is 0 on the left and 84 on the right.



In a similar way you can write an instruction and change the numbers on the dial to match any range you would like.


RUN this program and see that the knob rotates from 0 to 84.

RUN
GO




Now you can use the hand control to build your own video game. Begin with this portion of the program that makes a blinking target move around on the screen.

RESET
10X=RND(60)-31
20Y=RND(20)
30CLEAR
40X=X+RND(7)-4
50Y=Y+RND(7)-4
60BOX X,Y,4,4,1
70GOTO 30
GO
LIST
GO



First the computer picks an X between -30 and 30 and then a Y between 1 and 30. These values for X and Y are in the shaded area of the diagram below.



Lines 40 and 50 cause the target to wander around the screen. In line 40 the computer adds a random number to X. This moves the target to the right or left.


The number added to X is $RND(7) - 4$. $RND(7)$ is a random number between 1 and 7. Subtracting 4 makes this equal to a random number between -3 and 3.

In line 50 $RND(7) - 4$ is added to Y and this moves the target up or down.

The BOX is drawn at X and Y, and the program loops.

Now RUN the program and see that it puts a 4 x 4 black box somewhere in the shaded area.

RUN
GO



Now add a second box at the bottom of the screen. You will move this box left and right with the knob.

Notice that you will replace the old line 70 with a new instruction.




```
70K=KN(1)÷2
80BOX K,-40,3,8,1
90IF TR(1)=0GOTO 30
```

GO


LIST

GO



In line 70 the K counter is set to the value of the knob [KN(1)] divided by two.


Line 80 draws a black box that's three squares wide and eight squares tall. The box can be moved left or right as the K counter changes. The center of the box will be at -40, near the bottom of your screen.



When you pull the trigger, $TR(1)=1$. In line 90 the computer goes back to 30 if the trigger is not pulled and $TR(1)=0$. RUN the program and see if you can move the second box with the knob.

RUN


GO



Pull the trigger and see what happens and then LIST your program.

LIST

GO



When you pulled the trigger $TR(1)=1$. The computer did not go back to line 30 at the end of your program, it went on to the next instruction.

Now add the next instruction and tell the computer what to do when you pull the trigger.

100N=1

110IF K>X-3 IF K<X+3 N=15

120FOR A=1 TO N

130BOX K, 0, 1, 80, 3

140MU="V"


150NEXT A

160GOTO 10

GO

LIST

GO



Remember that the X counter moves the target left and right. The phaser at the bottom of the screen is moved left and right by the K counter. If $K=X$ when you pull the trigger, the laser and the target are lined up, and you've got a hit!

Hitting the target exactly is very hard, so line 110 allows a near miss to score. If K is within three dots either side of X, $N=15$.

The box in line 130 is eighty dots high and one dot wide. This forms the laser beam.


The N counter is set to 1 in line 100. If a hit is scored, $N=15$. Then the phaser fires N times in the FOR/NEXT loop. For a miss the beam fires once, and for a hit it fires fifteen times.

MU plays music like PRINT, but nothing is put on the screen.

After each shot the program loops back to the very beginning, puts a new target in a random location and then moves it around until you press the trigger again.

Now RUN your program and try your luck.

RUN
GO




This program could also be a two-player game if you use another hand control instead of the computer to move the target. Number 2, for example, is JX(2), JY(2), KN(2) and TR(2).

You could also add counters to keep and print the score, color the screen to show a hit, reverse the black and white for "night", and many other variations.

Programming Course

Lesson 8 Video Art

In this lesson you will learn how to use the power of your computer to create interesting and beautiful designs.



Here's a program that shows you all the colors in your computer and prints each color number.

RESET

10FOR A=0TO 255

20BC=A


30PRINT A

40NEXT A

GO

LIST

GO




The background color (BC) can be any number you select from 0 to 255. In this program the computer begins with color number 0 (Black) and shows each color and it's number.

Now RUN your program and see all the colors you can select from.

RUN

GO



Now enter this program and let the computer select the color while it draws random lines on your screen.

RESET

10BC=0

20CLEAR

30FC=RND (256)-1

40X=RND (160)-81

50Y=RND (88)-45


60LINE X,Y,3

70GOTO 30

GO

LIST


GO



First the computer sets the background color (BC) to black and clears the screen. In line 30 the foreground color (FC) is picked at random from the 256 possible choices. Then the computer draws a random line and goes back to instruction 30 to pick a new color and draws the next line.

RUN

GO



Now use the computer to draw a pattern of lines with this program. You will add colors later.


RESET

```
10INPUT A;CLEAR
20FOR N=79TO -79STEP -A
30LINE -N,43,1
40LINE N, -43,1
50NEXT N
60GOTO 10
```

GO

LIST

GO




The computer will ask you to input a number for A. This adjusts the spacing between the diagonal lines. Try a spacing of 3 for a start.

RUN

GO


3

GO




The computer is asking for a new A. Try a spacing of 5.

5
GO



Now try a spacing of 9.

9
GO



By just changing one number you have created three different designs. Now let the computer select the spacing. You must HALT the program before you can change it.


H
LIST
GO



Now make the spacing random with this new instruction.


10A=RND (12);CLEAR

GO
LIST
GO



RUN your program and let your computer change the design.

RUN
GO



Complete your design and color it with these additional instructions.

H

```
60FOR N=42TO -42STEP -A
70LINE 79,N,1
80LINE -79,-N,1
90NEXT N
100FOR A=1TO 500
110NEXT A
120CLEAR
130BC=RND (256)-1
140FC=BC+4+RND (32)×8
150GOTO 10
```



GO

LIST

GO


Lines 60, 70, 80, and 90 draw the second half of the design.

A slight pause is added in lines 100 and 110. This lets you see the pattern clearly before it changes again.

The background color is selected at random in line 130 and in the next line the foreground color is adjusted to match.

RUN

GO



COLOR WHEEL

Here's a color wheel you will use often because it helps you select colors and their numbers. Moving the number 1 hand control left and right selects the color. Moving it forward and backward selects the intensity. Pulling the trigger gives you a printout on the screen that shows that particular color number (0 to 31) color intensity (0 to 7) and the computer number (0 to 255). These numbers refer to the background color only. The foreground color is adjusted automatically so that you can read the numbers.

RESET

10C=+JX(1)

20IF C>31C=31

30IF C<0C=0

40I=I+JY(1)

50IF I>7I=7

60IF I<0I=0

70BC=C×8+I

80FC=BC+12

90IF TR(1)=0GOTO 10

100PRINT C,I,C×8+I

110GOTO 10

GO

LIST

GO

This program uses two counters, C and I to keep track of the color number and the intensity number. Both are adjusted by the hand control. JX(1) controls color and JY(1) controls intensity.

Lines 20 and 30 keep C between 0 and 31. Lines 50 and 60 keep I between 0 and 7.

The background color is set to the color number times eight plus the intensity number.

If the trigger is not pulled, the program loops back to line 10. Pulling the trigger prints the numbers in line 100 before looping back to line 10.

PROGRAMS

Here is an assortment of programs you can enter and run immediately. Pick a short program to begin with. If you have any difficulty return to the Introduction Section, page 4, for assistance.

If you make a mistake in punctuation, (as in leaving out a comma), the computer can not run your instruction. If this happens the computer will print the instruction on the screen with a question mark in the position of your error, to show you where your mistake is.

If you are using a program designed for one player be sure to use hand control number one, if it is a program for two players use hand controls numbers one and two only.

If at any time you wish to see your program, press LIST and your computer will show you what you have entered, up to that point.


You can change these programs any way you like. Change the instructions to make the computer do something different or add instructions to it. When you add instructions to your program, number the new line to fit between the existing lines. For example, if you want to add an instruction after line 30 and before line 40, number your instruction line 33 (or any number between 31 and 39).

Computer Games

PHASER PHUN

Try your skill as the computer moves the target. The first player's knob moves the phaser left or right and the trigger shoots.

```
1 .PHASER PHUN
2 .BY DICK AINSWORTH
10 X=RND (60)-31
20 Y=RND (20)
30 CLEAR
40 X=X+RND (7)-4
50 Y=Y+RND (7)-4
60 BOX X,Y,4,4,3
70 K=KN(1)/2
80 BOX K,-40,3,8,1
90 IF TR(1)=0GOTO 30
100 N=1
110 IF K>X-3 IF K<X+3N=15
120 FOR A=1TO N
130 BOX K,0,1,80,3
140 MU="4"
150 BC=A*8
160 NEXT A
170 FC=7
180 BC=8
```



You can make this a two-player game by changing these lines:

40X=X+JX(2)x3

50Y=Y+JY(2)x3


Player two controls the target while player one shoots.

ANTI-AIRCRAFT GUN

Player one moves the gun with the knob and shoots with the trigger.

Player two moves the plane right or left with JX(2) and controls the speed with the knob.

```
1 .ANTI-AIRCRAFT GUN
2 .BY BOB OGDON
10 CLEAR :W=-75;V=30;C=0
20 BC=22;FC=0
30 BOX -51,-30,5,5,1
40 D=KN(2)
50 IF D>50S=15;GOTO 80
60 IF D>-50S=10;GOTO 80
70 S=5
80 W=W+JX(2)*S
90 IF W>70GOTO 330
100 BOX 0,6,160,58,2
110 BOX W-4,34,2,1,1
120 BOX W,32,10,3,1
130 BOX W+5,32,1,1,1
140 IF TR(1)=1GOTO 260
150 P=KN(1)
160 IF P<-120X=-46;Y=-23;
GOTO 190
170 IF P<120X=-44;Y=-24;
GOTO 190
180 X=-43;Y=-25
190 LINE -48,-27,0
200 BOX -43,-23,10,10,2
210 LINE X,Y,1
220 IF X=-46U=-25
230 IF X=-44U=7
240 IF X=-43U=32
250 IF TR(1)=0GOTO 40
260 IF U=CGOTO 40
```



```


270 C=U
280 LINE U,V,1
290 NT=7;MU="V"
300 IF U-W<5IF U-W>-6GOTO 320
310 GOTO 40
320 GOSUB 400;GOTO 350
330 CX=-50;CY=0
340 PRINT "TOO BAD YOU MI
SSED"
350 IF TR(2)=1GOTO 10
360 GOTO 350
400 FOR Z=30TO -20STEP -2
0
410 BOX W+2,Z+4,1,2,1
420 BOX W,Z,3,10,1
430 BOX W,Z-6,1,1,1
440 BOX 0,6,160,58,2
450 NEXT Z
460 BC=74
470 FOR N=-5TO 5
480 LINE W,-25,0
490 LINE NXRND (5)+W,-25+
RND (10),3
500 MU=1
510 NEXT N
520 RETURN

```

ROCK/SHEAR/PAPER

Enter 1, 2, or 3 to select Rock, Shears, or Paper. Press GO and see if you beat the computer at this classic guessing game.

```
1 .ROCK/SHEARS/PAPER
2 .BY DICK AINSWORTH
10 H=0
20 C=0
30 GOSUB 301
40 GOSUB 302
50 GOSUB 303
60 PRINT "      HUMAN:",H
70 PRINT "COMPUTER:",C
80 A=RND (3)
90 INPUT "YOUR CHOICE:"B
100 PRINT "I PICKED ",
110 GOSUB 300+A
120 PRINT "YOU PICKED ",
130 GOSUB 300+B
140 IF A=BPRINT "A TIE!";
GOTO 30
150 IF A=1 IF B=3GOTO 240
160 IF A=2 IF B=1GOTO 240
170 IF A=3 IF B=2GOTO 240
180 PRINT "I WIN!"
190 NT=10
200 PRINT "135x105x10000"
210 NT=3
220 C=C+1
230 GOTO 30
240 PRINT "YOU WIN!"
250 NT=10
260 PRINT "3050034050000"
270 NT=3
```



COLOR WAR


One player tries to fill the screen with colored boxes, while the other tries to erase the pattern. The triggers are the secret. If your trigger is in the same position as your opponent's, the screen fills. If your trigger is in the opposite position, the pattern begins erasing itself. The two knobs control the colors of the pattern and background.

```
1 .COLOR WAR
2 .DICK AINSWORTH
10 CLEAR
20 BC=KN(1)+5*5
30 FC=KN(2)+5*5
40 X=RND (140)-80
50 Y=RND (70)-35
60 A=RND (25)
70 B=RND (25)
80 IF TR(1)=TR(2) C=1
90 IF TR(1)#TR(2) C=2
100 BOX X,Y,A,B,C
110 GOTO 20
```



SPIRAL

```
1 .SPIRAL1
2 .BY DICK AINSWORTH
10 S=RND (10);L=1;M=1
20 FOR N=79TO -79STEP -8
30 LINE -N,43,L
40 LINE N,-43,M
50 NEXT N
60 FOR N=42TO -42STEP -8
70 LINE 79,N,L
80 LINE -79,-N,M
90 NEXT N
100 FOR A=1TO 500
110 NEXT A
120 CLEAR
130 BC=RND (256)
140 FC=BC+4+RND (32)*8
150 GOTO 10
```



Electronic Music


COMPOSITION IN A

Enter the notes you wish to hear and then enter PRINT. The computer will play the first note; the first and second; the first, second, and third; and so on until it plays all the notes.

```
1 .COMPOSITION IN A
10 A=1
20 NT=5
30 K=KP
40 IF K=116GOTO 90
50 TV=K
60 @ (A)=K
70 A=A+1
80 GOTO 30
90 CLEAR
100 FOR N=1TO A-1
110 FOR P=1TO N
120 TV=@ (P)
130 NEXT P
140 CLEAR
150 NEXT N
160 GOTO 20
```

COMPOSITION A-Z

```
1 .COMPOSITION A-Z
10 CLEAR
20 A=RND (26)+64
30 MU=A
40 TV=A
50 GOTO 20
```



COMPOSITION IN F

Just enter the total number of notes, press GO and the computer will write and play a composition. Enter the number 15, for a start. The computer takes a while to work out the details, so you will have a short wait. Longer compositions can take several minutes to prepare.

```
1 .COMPOSITION 1/F
10 CLEAR :B=4:C=B*B
20 FOR D=1 TO B
30 @(D)=0
40 NEXT D
50 D=C+1
60 E=6
70 INPUT F
80 FOR G=0 TO D+F-1
90 H=A
100 A=A+1
110 I=A
120 J=C
130 K=0
140 FOR L=1 TO B
150 J=J+2
160 M=H+J
170 N=I+J
180 IF M<H H=H-J
190 IF N<I I=I-J
200 IF M=N GOTO 220
210 @(L)=RND (E)
220 K=K+@(L)
230 NEXT L
240 @(G)=K
250 IF A=C-1 A=0
260 NEXT G
270 INPUT NT
280 CLEAR
290 FOR L=0 TO D+F-1
300 TV= @(L)+"A"-B
310 NEXT L
320 NT=2
```

PLAYER PIANO

```
1 .PLAYER PIANO
2 .BY JAY FENTON
10 CLEAR
20 A=0
30 K=KP
40 IF K=116 GOTO 120
50 IF K=105 GOTO 10
60 IF K=30 A=A+1;GOTO 10
70 IF K=108 INPUT NT; GOTO 30
30
80 A=A+1
90 @ (A)=K
100 TV=K
110 GOTO 30
120 CLEAR
130 FOR C=1 TO A
140 TV=@ (C)
150 NEXT C
160 GOTO 30
```

See the electronic music section for complete details.
Your controls for this program are:

PRINT to play the notes you entered.

ERASE to back up and remove notes from the screen.

LINE to enter a new note time (Press GO after you enter the number.)

CLEAR to clear the notes from memory so that you can enter new music to be played.

PLAYER PIANO

Bagpipes

| | |
|------------------|------------------|
| 405654 | 46×2×164 |
| —70×2×106 | 606605 |
| 406654 | 46×2×164 |
| 502300 | 505505 |
| 405654 | 46×2×164 |
| —70×2×106 | 60×1×20×3 |
| ×406654 | ×4×2×1654 |
| 504401 | 605400 |

Melody

| | | |
|---------------|---------------|--------------|
| 506 | ×5×4×2 | ×300 |
| 70×1 | 76 6—6 | ×10×2 |
| ×400 | 506 | ×10×2 |
| ×300 | 70×1 | ×100 |
| ×300 | ×400 | 0 |
| ×200 | ×300 | ×500 |
| 600 | ×300 | ×500 |
| 000 | ×200 | ×200 |
| 70×1 | 600 | ×500 |
| +×10×2 | 70×5 | ×300 |
| ×700 | ×40 6 | ×600 |

PLAYER PIANO

March

| | | |
|--------|--------|--------|
| 5000 | +400+4 | 5000 |
| 034+4 | 50×10 | 50006 |
| 7000 | 0223 | 4000 |
| 3003 | 4000 | 0223 |
| 4070 | 600-6 | 5000 |
| 034+4 | 5000 | +400+4 |
| 5000 | 034+4 | 50×10 |
| ×200×1 | ×1000 | 0146 |
| ×1000 | 7006 | 5000 |
| 05×1×4 | ×3000 | ×300×2 |
| ×1000 | 0 | |

Marine's Hymn

| | | |
|-------|-------|-------|
| 13 | 5050 | 5050 |
| 500×1 | 5034 | 5050 |
| 4200 | 1000 | 0013 |
| 5050 | 5050 | 500×1 |
| 5034 | 5050 | 4200 |
| 1000 | 00×17 | 6040 |
| 6040 | 5006 | 50×17 |
| 6040 | 6×100 | 5000 |
| 0013 | 5050 | 5050 |
| 500×1 | 5034 | 5000 |
| 5000 | 6000 | 7000 |
| ×1000 | | |

PLAYER PIANO

Golden Slippers

■ = REST (USE SPACE KEY)

45

60606545

60606■45

6060656-7

60505■34

50505434

50505■34

50-7■6050

4000000■

10000■40

-70000000

-7060520■

30303040

50000■-70

6■6■5■5■

4000000■

40000■-70

×20×10-740■

-×30×20×150■

606060-70

×20×10-740■

50000■×10

×10000■60

6050410■

20000■50

-7060520■

30303040

50000■30

40304050

6000000■

10000■40

6050410■

20000■50

-7060-70×10

×2000000■

40000■-70

×20×10-740■

50000■×10

-×30×20×150■

606060-70

×10000■-×30

×20×20×10×10

PLAYER PIANO

Stars and Stripes Forever

5000

3000

5035

20+12

0032

2000

$\times 30 + \times 2 \times 3$

$\times 30 + \times 2 \times 3$

$\times 10 \times 10$

$\times 8000$

$\times 556 \times 3$

5043

00+23

4000

20+12

3500

00 \times 50

$\times 3000$

$\times 4 \times 3 \times 27$

$\times 107 \times 1$

0 \times 1 \times 2 \times 3

$\times 2000$

30+23

30+23

2002

4000

6060

$\times 50 \times 4 \times 3$

00+ $\times 2 \times 3$

$\times 2000$

$- \times 30 \times 2 \times 1$

$\times 5 \times 1 \times 2 \times 3$

$\times 1$

COMPOSITION IN L


First enter the number of the notes in the verse, then enter the notes to be played. Try entering a length number of 10. Press GO and about 20 notes as a start, then press PRINT. The computer will add one note and subtract one note, keeping the length constant.

```
1 .COMPOSITION IN L
10 A=1:INPUT L
20 NT=5
30 K=KP
40 IF K=116GOTO 90
50 TV=K
60 @ (A)=K
70 A=A+1
80 GOTO 30
90 CLEAR
100 FOR N=1TO A-1-L
110 FOR P=NT0 N+L
120 TV=@ (P)
130 NEXT P
140 CLEAR
150 NEXT N
```

Graphs and Charts


These programs draw line and bar graphs. Enter the number of items you wish to graph, then enter the value of each item.

```
1 .BAR GRAPH
5 CLEAR
10 INPUT "??"A
20 B=150/(A+1)
30 FOR N=1 TO A
40 PRINT N;
50 INPUT "? "@(N)
55 IF @(N)>87 GOTO 40
60 NEXT N
70 X=-80+B*2
80 CLEAR
100 FOR N=1 TO A
105 Y=@(N)*2-42
110 BOX X,Y,B*2,@(N),1
120 X=X+B
130 NEXT N
```



LINE GRAPH

```
1 .LINE GRAPH
5 CLEAR
10 INPUT "??"A
20 B=150/(A+1)
30 FOR N=1 TO A
40 PRINT N;
50 INPUT "? "@(N)
55 IF @(N)>87 GOTO 40
60 NEXT N
70 X=-80
80 CLEAR
90 LINE X,@(1)-44,0
100 FOR N=1 TO A
110 LINE X,@(N)-44,1
120 BOX X,-42,1,2,3
130 X=X+B
140 NEXT N
```



Video Art

COLOR WHEEL

Move hand control number one left or right to select the color, forward or backward to select the intensity. The trigger gives you a printout of the color (0 to 31), the intensity (0 to 7) and the color number (0 to 255).

```
1 .COLOR WHEEL
2 .DICK AIRSWORTH
10 C=0
20 H=0
30 C=C+JX(1)
40 IF C>31 C=31
50 IF C<0 C=0
60 H=H+JY(1)
70 IF H>7 H=7
80 IF H<0 H=0
90 BC=C*8+H
100 FC=BC+12
110 IF TR(1)=0 GOTO 30
120 PRINT C,H,C*8+H
130 GOTO 30
```

VIDEO WALLPAPER


```
1 .VIDEO WALLPAPER
2 .JAY FENTON
10 CLEAR
20 FOR A=1 TO 11
30 PRINT ;B=RND (8)+4
40 NEXT A
50 B=RND (8)+4
60 H=3
70 FOR C=1 TO RND (20)+8
80 A=RND (141)-71
90 BOX A,-40,B,8,H
100 NEXT C
110 PRINT
120 BOX CX,CY,6,8,2
130 GOTO 50
```



LASER DUEL


Two players cooperate or compete in forming designs as they each move one end of the reverse line.

```
1 .LASER DUEL
2 .BOB OGOON
10 CLEAR
20 X=0;Y=0
30 A=0;B=0
40 X=X+JX(1)*3
50 Y=Y+JY(1)*3
60 LINE X,Y,TR(1)+2
70 A=A+JX(2)*3
80 B=B+JY(2)*3
90 LINE A,B,TR(2)+2
100 BC=KN(1)+5*5
110 FC=KN(2)+5*5
120 GOTO 40
```




RND BOX 2

```
1 .RND BOX 2
10 CLEAR
20 B=20;E=3
30 BC=0;FC=22
40 BOX X,Y,B,B,E
50 BOX Y,X,B,B,E
60 BOX -X,Y,B,B,E
70 BOX Y,-X,B,B,E
80 BOX -X,-Y,B,B,E
90 BOX -Y,-X,B,B,E
100 BOX X,-Y,B,B,E
110 BOX -Y,X,B,B,E
120 X=RND (80)
130 Y=RND (80)
140 GOTO 20
```




PERSPECTIVE BOX

```
1 .PERSECTIVE BOX
2 .RICKEY SPIECE
10 CLEAR
20 A=60
30 X=40
40 Y=14
50 BC=0
60 FOR M=2TO A STEP 2
70 S=A+2-M
80 BOX X,Y,S,S,F
90 FC=RND (256)
100 X=X+1
110 Y=Y+1
120 IF F=-1 F=0;GOTO 140
130 F=-1
140 NEXT M
150 FOR S=2TO A STEP 2
160 X=X-1
170 Y=Y-1
180 BOX X,Y,S,S,0
190 NEXT S
200 GOTO 20
```



RANDOM LINE


```
1 .RANDOM LINE
2 .JEFF FREDERIKSEN
10 CLEAR
20 LINE RND (160)-80,RND
(88) -44,1
30 BC=RND (256)
40 FC=BC+132
50 GOTO 20
```



SCROLL ONE


These three programs are like paintings. The images evolve slowly and the visual experience changes over time.

```
1 .SCROLL ONE
2 .LARRY CUBA
10 BC=3
20 S=4+RND (4)
30 C=RND (S-3)
40 FOR A=-72TO 77STEP S
50 BOX A,-39,C,8,1
60 NEXT A
70 PRINT
80 FC=7+8×RND (32)
90 GOTO 10
```




SCROLL TWO

```
1 .SCROLL TWO
2 .LARRY CUBA
10 BC=3
20 S=4+RND (4)
30 C=RND (S-1)
40 FOR A=-72TO 77STEP S
50 T=RND (3)+1
60 FOR B=-43TO -36STEP T
70 BOX A,B,C,1,1
80 NEXT B
90 NEXT A
100 PRINT
110 FC=7+8×RND (32)
120 GOTO 10
```



SCROLL THREE


```
1 .SCROLL THREE
2 .LARRY CUBA
10 BC=0
20 S=4+RND (4)
30 U=1+RND (3)
40 C=RND (S-1)
50 FOR X=-72TO 77STEP U
60 BOX X,-39,1,8,1
70 NEXT X
80 FOR A=-72TO 77STEP S
90 T=RND (3)+1
100 FOR B=-43TO -36STEP T
110 BOX A,B,C,1,3
120 NEXT B
130 NEXT A
140 PRINT
150 FC=7+8×RND (32)
160 GOTO 10
```



SCRIBBLER



Player one controls the line on the screen. Direction and color of the line are changed by moving and rotating the knob. The trigger prints the line.

```
1 .SCRIBBLER
2 .BOB OGDON
10 CLEAR
20 BC=0
30 FC=KN(1)÷5×5
40 X=X+JX(1)×3
50 Y=Y+JY(1)×3
60 LINE X,Y,TR(1)
70 GOTO 30
```




ELECTRIC DOILY

```
1 .ELECTRIC DOILY
10 CLEAR
20 X=0;Y=0
30 X=X+RND (9)-5
40 Y=Y+RND (9)-5
50 A=2
60 B=3
70 C=3
80 M=X
90 N=Y
100 GOSUB 1000
110 M=-X
120 GOSUB 1000
130 N=-Y
140 GOSUB 1000
150 M=X
160 GOSUB 1000
170 N=Y
180 M=X
190 GOSUB 1000
200 N=-Y
210 GOSUB 1000
220 M=-X
230 GOSUB 1000
240 N=Y
250 GOSUB 1000
260 IF X>50 GOTO 20
270 IF Y>50 GOTO 20
280 GOTO 30
1000 BOX M,N,A,B,C
1010 RETURN
```



COLORING BOX

```
1 .COLORING BOX
2 .BY JEFF FREDERIKSEN
10 CLEAR
20 A=RND (40)*2
30 BOX 0,0,A,A,3
40 IF BC>256 BC=-1
50 BC=BC+8
60 GOTO 20
```



RUBBER BAND

Draw a connect-the-dots pattern on the screen. Moving the knob controls direction of the line. Rotating the knob to the right draws a line and rotating the knob to the left leaves a space. The trigger prints each section of the line.

```
1'.RUBBER BANDS
2'.JAY FENTON
10 CLEAR
20 A=0
30 CLEAR
40 B=0
50 C=0
60 D=0
70 A=A+JX(1)
80 B=B+JY(1)
90 LINE C,D,0
100 LINE A,B,3
110 IF TR(1)GOTO 160
120 LINE C,D,0
130 LINE A,B,3
140 IF KN(1)<0GOTO 170
150 GOTO 70
160 LINE C,D,0;LINE A,B,1
170 C=A
180 D=B
190 IF TR(1)GOTO 190
200 GOTO 70
```

Learning Skills

LETTER MATCH

This learning program becomes easier or more difficult, to match the player's skill.


```
1 .LETTER MATCH
10 S=4
15 CLEAR
20 PRINT "CAN YOU REMEMB
ER"
30 PRINT S," LETTERS?"
40 FOR N=1 TO S
50 @ (N)=RND (26)+64
60 TV=@ (N)
70 NEXT N
80 GOSUB 1000
90 CLEAR
100 PRINT "GO!"
110 FOR N=1 TO S
120 G=KP
130 TV=G
140 IF G#@ (N) GOTO 200
150 NEXT N: S=S+1
160 GOSUB 1000
170 GOTO 15
200 PRINT "SORRY"
210 FOR N=1 TO S
220 TV=@ (N)
230 NEXT N
240 GOSUB 1000
250 S=S-1
260 GOTO 15
1000 FOR T=1 TO 500
1010 NEXT T
1020 RETURN
```


MATH QUIZ

```

1 .MATH QUIZ
10 A=RND (10)
20 B=RND (10)
30 S=RND (4)
40 IF S=1GOTO 110
50 IF S=2GOTO 150
60 IF S=3GOTO 190
70 PRINT A;#1,"+",B,"=",
80 INPUT "?"C
90 IF C=A+BPRINT "RIGHT!"
";GOTO 10
100 GOTO 70
110 PRINT A;#1,"-",B,"=",
120 INPUT "?"C
130 IF C=A-BPRINT "CORREC
T!";GOTO 10
140 GOTO 110
150 PRINT A;#1,"x",B,"=",
160 INPUT "?"C
170 IF C=AxBPRINT "GOOD!"
;GOTO 10
180 GOTO 150
190 PRINT AxB;#1,"÷",A,"=
."
200 INPUT "?"C
210 IF C=BPRINT "TERRIFIC
!";GOTO 10
220 GOTO 190

```




NUMBER MATCH

```

1 .NUMBER MATCH
10 CLEAR
20 A=RND (10)
30 INPUT "YOUR GUESS:"B
40 IF A=B GOTO 80
50 IF A>B PRINT "MORE↑"
60 IF A<B PRINT "LESS↓"
70 GOTO 30
80 PRINT B," IS RIGHT!"
90 GOTO 10

```



Terms and Symbols

Computer Words

BOX X,Y,A,B,1

means draw a black box that's centered at the point X,Y. The box is A dots wide and B dots high. You can draw:

BOX X,Y,A,B,1 black box

BOX X,Y,A,B,2 white box

BOX X,Y,A,B,3 reverse box

BOX X,Y,A,B,4 no box

CLEAR

means clear the screen.

ERASE

means forget the last key you pushed. This doesn't work if the last key was RUN, H, or GO.

FOR/TO/NEXT/STEP

These words all work together to make a loop.

```
10FOR A=1TO 16STEP 3
```

```
20PRINT A,
```

```
30NEXT A
```

This loop prints 1, 4, 7, 10, 13, 16.

GO

means go. Press GO after each instruction.

GO + 10

means go to the next line and add 10 to the line number.

GOTO 20

means go to line number 20 and continue running the program.

GOSUB 200

means go to line number 200 and continue running the program until the word RETURN, then return to the instruction that follows GOSUB 200.



means halt the program and return control to you.

IF

means check and see whether something is true or not. IF A=5 GOTO 20 means if the number in the A counter is 5, go to line 20; if it isn't 5 then go to the next instruction.

INPUT A

means stop and wait for you to enter a number which is put into the A counter when you press GO.

INPUT "HOW MANY?" A

means print "HOW MANY?" on the screen and then input a number for the A counter.

LINE X,Y,1

means draw a black line on the screen to the point X,Y.

You can draw:

LINE X,Y,1 black line

LINE X,Y,2 white line

LINE X,Y,3 reverse line

LINE X,Y,4 no line

LIST

means print on the TV all the instructions now in the computer after you press GO.

LIST 100

means start with line number 100 and list.

LIST 100,5

means start with line 100 and list the next five instructions.

PAUSE

means stop the computer. You can pause while running or listing a program. Press any key to continue.

PRINT "A"

means print the character A on the screen.

PRINT A

print the value of the A counter on the screen.

SZ

is the number of unused memory locations.

PX(X,Y)

is the dot at location X. 1 is black, 0 is white.

RETURN

means return to the line following the word GOSUB. The computer remembers which GOSUB to return to.

RND (A)

means pick a random number between one and the number in the A counter.

RND (5)

means pick a random number between one and five.

RUN

means run the program after you press GO.

SPACE

means leave a space on the screen. Spaces don't matter to the computer, but they can make your instructions easier for you to read.

Inputs, Outputs, and Controls

JX(1)

is a number that matches the position of the number one hand control.

Left $JX(1) = -1$

Center $JX(1) = 0$

Right $JX(1) = 1$

JY(1)

is a number that matches the position of the number one hand control.

Forward $JY(1) = 1$

Center $JY(1) = 0$

Back $JY(1) = -1$

TR(1)

is a number that matches the trigger on the number one hand control.

Pulled $TR(1) = 1$

Not Pulled $TR(1) = 0$

KN(1)

is a number that matches the position of the knob on hand control number one.

A=KP

means wait until you press a key on the keypad.

Each key has a number and the number of the key you press is stored in the A counter. You can see what key you pressed with the instruction, $TV=A$.

TV=A

means put a letter or other character on the TV.
The character is the one that matches the number in the A counter. See KP.

MU=A

means play a note in the TV speaker that matches the number in A counter.

MU="A"

means play a note in the TV speaker that's the same as the note you hear when you press the letter A.

FC

is the number of the foreground color.

BC

is the number of the background color.

NT

is the note time. After RESET the note time is set at three. Note times are slower if NT is larger than three.

CX

is the number that places the cursor (black square) left or right.

CY

is the number that places the cursor (black square) up or down.

PRINT #A,B

means leave A spaces and then print the number in the B counter.

, The comma means continue. In PRINT A, the comma after A means continue printing on the same line.

; The semi-colon means the same thing as a line number.

```
10PRINT A; GOTO 30
```

is the same as

```
10PRINT A
```

```
20GOTO 30
```

You can use the semi-colon to put two or more instructions on the same line.

>

means "is greater than," as $5 > 3$.

<

means "is less than," as in $8 < 12$.

=

means "is equal to."

A=5

means put the number 5 in the A counter.

#

means "not equal to."

Control Words

The following control words are used with the Bally Audio Tape Interface Accessory.

:PRINT

means tape print on record data from memory.

:INPUT

means tape input or playback data from tape to memory.

:LIST

means tape list or playback data on the screen.

:RETURN

means tape return or end record or playback mode.

Error Messages

WHAT?

The computer says WHAT? when it doesn't understand you.

HOW?

The computer asks HOW? when it understands what you want but can not figure out how to do it.

SORRY!

The computer says SORRY! when there isn't enough room in its memory to do what you want.

Arithmetic

Your computer is designed to work the multiplication and division portions of a problem first, and the addition and subtraction portions last.

$$3 \times 5 - 2 = 13 \text{ (not 9)}$$

Parenthesis will change this order.

$$3 \times (5 - 2) = 9 \text{ (not 13)}$$

Whole numbers only are used.

$$15 \div 2 = 7 \text{ (not } 7\frac{1}{2}\text{)}$$

**This Page Intentionally Left Blank
(Inside, Back Cover)**

**Bally Consumer Products Division
Bally Manufacturing Corporation
10750 West Grand Ave., Franklin Park, Ill. 60131
PRINTED IN U.S.A.**